```
ClearAll[x1, x2, θ1, θ2, t]
T1 = 1/2 m1 (x1'[t]^2 + ((L + x1[t]) θ1'[t])^2);
V1 = -m1 g (L + x1[t]) Cos[θ1[t]] + 1/2 k1 x1[t]^2;
T2 =
   1/2 m2 ((x2'[t] + x1'[t] Cos[θ1[t] - θ2[t]])^2 + (x1'[t] Sin[θ1[t] - θ2[t]])^2) +
     1/2 m2 (((L + x2[t]) θ2'[t] + ((L + x1[t]) θ1'[t] Cos[θ1[t] - θ2[t]]))^2 +
          ((L + x1[t]) θ1'[t] Sin[θ1[t] - θ2[t]])^2);
V2 = -m2 g ((L + x1[t]) Cos[θ1[t]] + (L + x2[t]) Cos[θ2[t]]) + 1/2 k2 x2[t]^2;


(*Lagrangian*)

In[71]:= (lag = (T1 + T2) - (V1 + V2)) // Simplify
```

Out[71]=

$$\frac{1}{2}\Big(-k1\,x1[t]^2 + 2\,g\,m1\,Cos[\theta1[t]]\,(L + x1[t]) - k2\,x2[t]^2 + 2\,g\,m2\,(Cos[\theta1[t]]\,(L + x1[t]) + Cos[\theta2[t]]\,(L + x2[t])) + m2\,(x1'[t]^2 + 2\,Cos[\theta1[t] - \theta2[t]]\,x1'[t]\,x2'[t] + x2'[t]^2) + m1\,(x1'[t]^2 + (L + x1[t])^2\,\theta1'[t]^2) + m2\,\big(Sin[\theta1[t] - \theta2[t]]^2\,(L + x1[t])^2\,\theta1'[t]^2 + (Cos[\theta1[t] - \theta2[t]]\,(L + x1[t])\,\theta1'[t] + (L + x2[t])\,\theta2'[t])^2\big)\Big)$$

```
(*x1 *)

In[72]:= (eq1 = D[D[lag, x1'[t]], t] - D[lag, x1[t]] == 0) // Simplify

(*x2*)

In[73]:= (eq2 = D[D[lag, x2'[t]], t] - D[lag, x2[t]] == 0) // Simplify

(*theta 1*)

In[75]:= (eq3 = D[D[lag, θ1'[t]], t] - D[lag, θ1[t]] == 0) // Simplify

In[76]:= (eq4 = D[D[lag, θ2'[t]], t] - D[lag, θ2[t]] == 0) // Simplify

(*Numerically solve the equations of motion*)

pars = {L -> 1, m1 → 1, m2 → 2, g → 9.8, k1 → 10, k2 → 30};
ic = {θ1[0] == 5 Degree , θ1'[0] == 0, θ2[0] == 3 Degree,
      θ2'[0] == 0, x1[0] == 0, x1'[0] == 0, x2[0] == 0, x2'[0] == 0};
eqs = Flatten[{eq1, eq2, eq3, eq4}] /. pars

In[122]:= numericalSolution = First@NDSolve[{eqs, ic}, {x1, x2, θ1, θ2}, {t, 0, 20}];
```

In[120]:=
```
Plot[Evaluate[({θ1[t], θ2[t]} /. numericalSolution) * 180 / Pi],
  {t, 0, 20}, PlotRange → All, AxesLabel → {"time", "Angle (in degree)"},
  ImageSize → 400, PlotLegends → {"mass 1", "mass 2"}]
```

Out[120]=



In[121]:=
```
Plot[Evaluate[{x1[t], x2[t]} /. numericalSolution], {t, 0, 20},
  PlotRange → All, AxesLabel → {"time", "spring extensions in meters"},
  ImageSize → 400, PlotLegends → {"mass 1", "mass 2"}]
```

Out[121]=